In[ ]:=
```
(*For complicated group and algebraic problems,
one can use GAP https://www.gap-system.org
   Here we define the group algebra using a generic non-
  cummutative algebra structure*)
```

In[1]:=
```
Needs["NC`"]
Needs["NCAlgebra`"]
SetOptions[inv, Distribute → True]
NCSetOutput[NonCommutativeMultiply → False]
```

In[5]:=
```
DnRelations[n_] :=
 {r^n → 1, s^2 → 1, (r ** s)^2 → 1, s ** r → r^(n-1) ** s(*Define a preferred order*)}
D4Relations = DnRelations[4];
```

In[7]:=
```
D4 = Flatten[Table[r^i ** s^j, {i, 0, 3}, {j, 0, 1}]]
```

Out[7]=
```
{1, s, r, r ** s, r ** r, r ** r ** s, r ** r ** r, r ** r ** r ** s}
```

In[8]:=
```
a_ ∘ b_ := NCReplaceRepeated[NCExpand[a ** b], D4Relations]
a__ ∘ b_ ∘ c_ := a ∘ (b ∘ c)
```

In[16]:=
```
conjugateQ[a_, b_, G__] := Or @@ (a == # ∘ b ∘ inv[#] & /@ G)
conjugates = Select[Position[
    Table[conjugateQ[D4[[i]], D4[[j]], D4], {i, 8}, {j, 8}], True], #[[1]] < #[[2]] &];
conjClass = Join[
  List /@ DeleteCases[Range[8], Alternatives @@ Flatten[conjugates]], conjugates]
classOps = Total[D4[[#]] & /@ #] & /@ conjClass;
TableForm[classOps, TableHeadings → {Table["C" <> ToString[i], {i, 5}], None}]
```

Out[18]=
```
{{1}, {5}, {2, 6}, {3, 7}, {4, 8}}
```

Out[20]//TableForm=

| | |
|---|---|
| C1 | 1 |
| C2 | r ** r |
| C3 | s + r ** r ** s |
| C4 | r + r ** r ** r |
| C5 | r ** s + r ** r ** r ** s |

In[21]:=
```
MatrixForm@
 (classMultiTable = Simplify[Table[classOps[[i]] ∘ classOps[[j]], {i, 5}, {j, 5}] /.
     {r^3 ** s → C5 - r ** s, r^3 → C4 - r, r^2 ** s → C3 - s, r^2 → C2}] /. {Id → C1})
```

Out[21]//MatrixForm=

$$\begin{pmatrix} C_1 & C_2 & C_3 & C_4 & C_5 \\ C_2 & C_1 & C_3 & C_4 & C_5 \\ C_3 & C_3 & 2\,(C_1+C_2) & 2\,C_5 & 2\,C_4 \\ C_4 & C_4 & 2\,C_5 & 2\,(C_1+C_2) & 2\,C_3 \\ C_5 & C_5 & 2\,C_4 & 2\,C_3 & 2\,(C_1+C_2) \end{pmatrix}$$

In[22]:=
```
(*C^k_{i,j}=:c[i,j,k]*)
coeff[i_, j_, k_] :=
 classMultiTable[[i, j]] /. (C_# → KroneckerDelta[#, k] & /@ Range[5])
```

In[23]:=
```
coefftable = Table[coeff[i, j, k], {i, 5}, {j, 5}, {k, 5}]
```

Out[23]=
```
{{{1, 0, 0, 0, 0}, {0, 1, 0, 0, 0}, {0, 0, 1, 0, 0}, {0, 0, 0, 1, 0}, {0, 0, 0, 0, 1}},
 {{0, 1, 0, 0, 0}, {1, 0, 0, 0, 0}, {0, 0, 1, 0, 0}, {0, 0, 0, 1, 0}, {0, 0, 0, 0, 1}},
 {{0, 0, 1, 0, 0}, {0, 0, 1, 0, 0}, {2, 2, 0, 0, 0}, {0, 0, 0, 0, 2}, {0, 0, 0, 2, 0}},
 {{0, 0, 0, 1, 0}, {0, 0, 0, 1, 0}, {0, 0, 0, 0, 2}, {2, 2, 0, 0, 0}, {0, 0, 2, 0, 0}},
 {{0, 0, 0, 0, 1}, {0, 0, 0, 0, 1}, {0, 0, 0, 2, 0}, {0, 0, 2, 0, 0}, {2, 2, 0, 0, 0}}}
```

In[24]:=
```
ClearAll[Lmat]
Lmat[j_, k_] := Sum[coefftable[[i, j, k]] × Y[i], {i, 5}]
MatrixForm@(LmatVal = Table[Lmat[j, k], {j, 5}, {k, 5}])
```

Out[26]//MatrixForm=

$$\begin{pmatrix} Y[1] & Y[2] & Y[3] & Y[4] & Y[5] \\ Y[2] & Y[1] & Y[3] & Y[4] & Y[5] \\ 2\,Y[3] & 2\,Y[3] & Y[1]+Y[2] & 2\,Y[5] & 2\,Y[4] \\ 2\,Y[4] & 2\,Y[4] & 2\,Y[5] & Y[1]+Y[2] & 2\,Y[3] \\ 2\,Y[5] & 2\,Y[5] & 2\,Y[4] & 2\,Y[3] & Y[1]+Y[2] \end{pmatrix}$$

In[27]:=
```
{evals, efuns} = Eigensystem[LmatVal];
TableForm@evals
MatrixForm@efuns
```

Out[28]//TableForm=

$$Y[1]-Y[2]$$
$$Y[1]+Y[2]+2\,Y[3]-2\,Y[4]-2\,Y[5]$$
$$Y[1]+Y[2]-2\,Y[3]+2\,Y[4]-2\,Y[5]$$
$$Y[1]+Y[2]-2\,Y[3]-2\,Y[4]+2\,Y[5]$$
$$Y[1]+Y[2]+2\,Y[3]+2\,Y[4]+2\,Y[5]$$

Out[29]//MatrixForm=

$$\begin{pmatrix} -1 & 1 & 0 & 0 & 0 \\ -\frac{1}{2} & -\frac{1}{2} & -1 & 1 & 1 \\ -\frac{1}{2} & -\frac{1}{2} & 1 & -1 & 1 \\ \frac{1}{2} & \frac{1}{2} & -1 & -1 & 1 \\ \frac{1}{2} & \frac{1}{2} & 1 & 1 & 1 \end{pmatrix}$$